/*
 * Copyright the original author or authors.
 * 
 * Licensed under the MOZILLA PUBLIC LICENSE, Version 1.1 (the "License");
 * you may not use this file except in compliance with the License.
 * You may obtain a copy of the License at
 * 
 *	  http://www.mozilla.org/MPL/MPL-1.1.html
 * 
 * Unless required by applicable law or agreed to in writing, software
 * distributed under the License is distributed on an "AS IS" BASIS,
 * WITHOUT WARRANTIES OR CONDITIONS OF ANY KIND, either express or implied.
 * See the License for the specific language governing permissions and
 * limitations under the License.
 */
package com.googlecode.milib.data {
	import com.googlecode.milib.base.MLBroadcastClass;

	/**
	 * @author Marek Brun
	 */
	public class BrowseHistory extends MLBroadcastClass {

		public const event_Changed : String = 'Changed';
		
		private var history:Array;
		private var backCount:uint;

		public function BrowseHistory() {
			history=[];
			backCount=0;
		}
		
		public function add(historyStepID:*):void {
			history.splice(history.length-backCount);
			history.push(historyStepID);
			backCount=0;
			broadcastEvent(event_Changed);
		}
		
		public function back():void {
			if(!getCanGoBack()){ return; }
			backCount++;
			broadcastEvent(event_Changed);
		}
		
		public function forward():void {
			if(!getCanGoForward()){ return; }
			backCount--;
			broadcastEvent(event_Changed);
		}
		
		public function getCanGoForward():Boolean {
			return backCount>0;
		}
		
		public function getCanGoBack():Boolean {
			return Boolean(history[history.length-1-backCount-1]);
		}
		
		public function getCurrentHistoryStepID():* {
			return history[history.length-1-backCount];
		}
		
	}
}
